NAG Library Routine Document

S30QCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S30QCF computes the Bjerksund and Stensland (2002) approximation to the price of an American option.

2 Specification

```
SUBROUTINE S30QCF (CALPUT, M, N, X, S, T, SIGMA, R, Q, P, LDP, IFAIL)
INTEGER             M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, T(N), SIGMA, R, Q, P(LDP,N)
CHARACTER(1)        CALPUT
```

3 Description

S30QCF computes the price of an American option using the closed form approximation of Bjerksund and Stensland (2002). The time to maturity, T, is divided into two periods, each with a flat early exercise boundary, by choosing a time $t \in [0,T]$, such that $t = \frac{1}{2}(\sqrt{5}-1)T$. The two boundary values are defined as $\tilde{x} = \tilde{X}(t)$, $\tilde{X} = \tilde{X}(T)$ with

$$\tilde{X}(\tau) = B_0 + (B_\infty - B_0)(1 - \exp\{h(\tau)\}),$$

where

$$h(\tau) = -\big(b\tau + 2\sigma\sqrt{\tau}\big)\left(\frac{X^2}{(B_\infty - B_0)B_0}\right),$$

$$B_\infty \equiv \frac{\beta}{\beta - 1}X, \quad B_0 \equiv \max\Big\{X, \Big(\frac{r}{r-b}\Big)X\Big\},$$

$$\beta = \left(\frac{1}{2} - \frac{\mathrm{b}}{\sigma^2}\right) + \sqrt{\left(\frac{b}{\sigma^2} - \frac{1}{2}\right)^2 + 2\frac{r}{\sigma^2}}.$$

with $b = r - q$, the cost of carry, where r is the risk-free interest rate and q is the annual dividend rate. Here X is the strike price and σ is the annual volatility.

The price of an American call option is approximated as

$$\begin{aligned} P_{\text{call}} \;=\; & \alpha\big(\tilde{X}\big)S^\beta - \alpha\big(\tilde{X}\big)\phi\big(S,t|\beta,\tilde{X},\tilde{X}\big) + \\ & \phi\big(S,t|1,\tilde{X},\tilde{X}\big) - \phi\big(S,t|1,\tilde{x},\tilde{X}\big) - \\ & X\phi\big(S,t|0,\tilde{X},\tilde{X}\big) + X\phi\big(S,t|0,\tilde{x},\tilde{X}\big) + \\ & \alpha(\tilde{x})\phi\big(S,t|\beta,\tilde{x},\tilde{X}\big) - \alpha(\tilde{x})\Psi\big(S,T|\beta,\tilde{x},\tilde{X},\tilde{x},t\big) + \\ & \Psi\big(S,T|1,\tilde{x},\tilde{X},\tilde{x},t\big) - \Psi\big(S,T|1,X,\tilde{X},\tilde{x},t\big) - \\ & X\Psi\big(S,T|0,\tilde{x},\tilde{X},\tilde{x},t\big) + X\Psi\big(S,T|0,X,\tilde{X},\tilde{x},t\big), \end{aligned}$$

where α, ϕ and Ψ are as defined in Bjerksund and Stensland (2002).

The price of a put option is obtained by the put-call transformation,

$$P_{\text{put}}(X,S,T,\sigma,r,q) = P_{\text{call}}(S,X,T,\sigma,q,r).$$

The option price $P_{ij} = P\big(X = X_i, T = T_j\big)$ is computed for each strike price in a set X_i, $i = 1,2,\ldots,m$, and for each expiry time in a set T_j, $j = 1,2,\ldots,n$.

4 References

Bjerksund P and Stensland G (2002) Closed form valuation of American options **Discussion Paper 2002/09** *NHH Bergen Norway* http://www.nhh.no/

Genz A (2004) Numerical computation of rectangular bivariate and trivariate Normal and t probabilities *Statistics and Computing* **14** 151–160

5 Arguments

1: CALPUT – CHARACTER(1) *Input*

On entry: determines whether the option is a call or a put.

CALPUT = 'C'
A call; the holder has a right to buy.

CALPUT = 'P'
A put; the holder has a right to sell.

Constraint: CALPUT = 'C' or 'P'.

2: M – INTEGER *Input*

On entry: the number of strike prices to be used.

Constraint: M ≥ 1.

3: N – INTEGER *Input*

On entry: the number of times to expiry to be used.

Constraint: N ≥ 1.

4: X(M) – REAL (KIND=nag_wp) array *Input*

On entry: X(i) must contain X_i, the ith strike price, for $i = 1, 2, \ldots, \mathrm{M}$.

Constraint: X(i) $\geq z$ and X(i) $\leq 1/z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots, \mathrm{M}$.

5: S – REAL (KIND=nag_wp) *Input*

On entry: S, the price of the underlying asset.

Constraint: S $\geq z$ and S $\leq \frac{1}{z}$, where $z =$ X02AMF(), the safe range parameter and S$^{\beta} < \frac{1}{z}$ where β is as defined in Section 3.

6: T(N) – REAL (KIND=nag_wp) array *Input*

On entry: T(i) must contain T_i, the ith time, in years, to expiry, for $i = 1, 2, \ldots, \mathrm{N}$.

Constraint: T(i) $\geq z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots, \mathrm{N}$.

7: SIGMA – REAL (KIND=nag_wp) *Input*

On entry: σ, the volatility of the underlying asset. Note that a rate of 15% should be entered as 0.15.

Constraint: SIGMA > 0.0.

8: R – REAL (KIND=nag_wp) *Input*

On entry: r, the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.

Constraint: R ≥ 0.0.

9: Q – REAL (KIND=nag_wp) *Input*

On entry: q, the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.

Constraint: Q ≥ 0.0.

10: P(LDP, N) – REAL (KIND=nag_wp) array *Output*

On exit: P(i,j) contains P_{ij}, the option price evaluated for the strike price X_i at expiry T_j for $i = 1, 2, \ldots, \mathrm{M}$ and $j = 1, 2, \ldots, \mathrm{N}$.

11: LDP – INTEGER *Input*

On entry: the first dimension of the array P as declared in the (sub)program from which S30QCF is called.

Constraint: LDP $\geq$ M.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, CALPUT $= \langle value \rangle$ was an illegal value.

IFAIL $= 2$

On entry, M $= \langle value \rangle$.
Constraint: M ≥ 1.

IFAIL $= 3$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 1.

IFAIL $= 4$

On entry, X$(\langle value \rangle) = \langle value \rangle$.
Constraint: X$(i) \geq \langle value \rangle$ and X$(i) \leq \langle value \rangle$.

IFAIL $= 5$

On entry, S $= \langle value \rangle$.
Constraint: S $\geq \langle value \rangle$ and S $\leq \langle value \rangle$.

IFAIL $= 6$

On entry, T($\langle value \rangle$) $= \langle value \rangle$.
Constraint: T(i) $\geq \langle value \rangle$.

IFAIL $= 7$

On entry, SIGMA $= \langle value \rangle$.
Constraint: SIGMA > 0.0.

IFAIL $= 8$

On entry, R $= \langle value \rangle$.
Constraint: R ≥ 0.0.

IFAIL $= 9$

On entry, Q $= \langle value \rangle$.
Constraint: Q ≥ 0.0.

IFAIL $= 11$

On entry, LDP $= \langle value \rangle$ and M $= \langle value \rangle$.
Constraint: LDP $\geq$ M.

IFAIL $= 14$

On entry, S $= \langle value \rangle$ and $\beta = \langle value \rangle$.
Constraint: S$^{\beta} < \langle value \rangle$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The accuracy of the output will be bounded by the accuracy of the cumulative bivariate Normal distribution function. The algorithm of Genz (2004) is used, as described in the document for G01HAF, giving a maximum absolute error of less than 5×10^{-16}. The univariate cumulative Normal distribution function also forms part of the evaluation (see S15ABF and S15ADF).

8 Parallelism and Performance

S30QCF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example computes the price of an American call with a time to expiry of 3 months, a stock price of 110 and a strike price of 100. The risk-free interest rate is 8% per year, there is an annual dividend return of 12% and the volatility is 20% per year.

10.1 Program Text

```
    Program s30qcfe

!     S30QCF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s30qcf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: q, r, s, sigma
      Integer                          :: i, ifail, j, ldp, m, n
      Character (1)                    :: calput
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: p(:,:), t(:), x(:)
!     .. Executable Statements ..
      Write (nout,*) 'S30QCF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) calput
      Read (nin,*) s, sigma, r, q
      Read (nin,*) m, n

      ldp = m
      Allocate (p(ldp,n),t(n),x(m))

      Read (nin,*)(x(i),i=1,m)
      Read (nin,*)(t(i),i=1,n)

      ifail = 0
      Call s30qcf(calput,m,n,x,s,t,sigma,r,q,p,ldp,ifail)

      Write (nout,*)

      Select Case (calput)
      Case ('C','c')
        Write (nout,*) 'American Call :'
      Case ('P','p')
        Write (nout,*) 'American Put :'
      End Select

      Write (nout,99998) '  Spot       = ', s
      Write (nout,99998) '  Volatility = ', sigma
      Write (nout,99998) '  Rate       = ', r
      Write (nout,99998) '  Dividend   = ', q

      Write (nout,*)
      Write (nout,*) '   Strike    Expiry    Option Price'

      Do i = 1, m
```

```
        Do j = 1, n
          Write (nout,99999) x(i), t(j), p(i,j)
        End Do

      End Do

99999 Format (1X,2(F9.4,1X),6X,F9.4)
99998 Format (A,1X,F8.4)
    End Program s30qcfe
```

10.2 Program Data

```
S30QCF Example Program Data
 'C'                 : Call = 'C', Put = 'P'
 110.0 0.2 0.08 0.12 : S, SIGMA, R, Q
 1 1                 : M, N
 100.0               : X(I), I = 1,2,...M
 0.25                : T(I), I = 1,2,...N
```

10.3 Program Results

```
 S30QCF Example Program Results

 American Call :
  Spot       =  110.0000
  Volatility =    0.2000
  Rate       =    0.0800
  Dividend   =    0.1200

    Strike    Expiry    Option Price
  100.0000    0.2500         10.3340
```